
第１章 UARDECSの活用 

１ はじめに 

Arduinoは世界中で使用されているマイコンボードで PCとUSB接続してプログラムを書き換える

ことで様々な用途に活用できるものです。日本でも概ね 2000～6000円の価格帯で販売されており、

通販などで容易に入手できます。Arduinoの開発環境は無料で公開されており、C言語に似たスケ

ッチと呼ばれる言語で記述でき、使い方に関しては無数の文献が web上で公開されています。今

回は、これを利用して UECS対応ノードを自作する方法を紹介します。 

 

２ Arduino と Raspberry Pi 

 両方とも大変普及したマイコンボードですが、Arduinoは Raspberry Pi より単純な構造をしていま

す。一方、Raspberry Piはパソコンに準じる機能を持ち、ソフトウェアを重視した開発に向いていま

す。スペックでは Raspberry Piの方が Arduino より優れていますが、OSを動作させるオーバーヘッ

ドが生じるため、センサなどの入力に高速に応答するにはArduinoが向きます。ArduinoはCPU内

部のフラッシュメモリに書き込まれたプログラムを実行しますので OSの書かれた SDカードからプロ

グラムを読み込んで動作する Raspberry Pi より高速に起動し、プログラム記録メディアの寿命を気

にする必要がないという利点があります。また、Arduinoは回路が公開され、独自に設計した互換

機を合法的に製造できます。一方、Raspberry Piは大容量なメモリと SDカードを利用して複雑なプ

ログラムを実行でき、ネットワーク関係のソフトウェアも充実しているので相互補完的に利用するの

が良いと思われます。Raspberry Pi用の UECS対応開発環境には UECS-Piがあります。 

 

３ UARDECS とは 

 UARDECSは Arduino用に UECS対応ノードの開発を支援するためのミドルウェアです。これは、

安場氏が開発したものを筆者(黒崎)が改修したもので、以下のサイトから入手できます。本稿執筆

時点で最新版は 0.8です。 

UECS ノード開発用ミドルウェア(http://uecs.org/arduino/uardecs.html) 

このミドルウェアは、UECSの通信実用規約"1.00-E10" に対応した通信文の送受信管理とマイ

コンボードに搭載した httpサーバを利用した設定画面の生成、設定値の不揮発性メモリへの自動

読み書き等を行うことができます。インストールの方法はマニュアルがあるので割愛させて頂きます

が、本稿執筆時点で Arduino IDE 1.8.0（配布元: http://www.arduino.org/）がリリースされており、

これに対応しています。UARDECSは発展途上のソフトウェアであり、今後も改修を行っていく予定

です。 

http://www.wa-bit.com/business_info-2/smart-agri-project/smart-agri-products/uecs-pi
http://uecs.org/arduino/uardecs.html
http://www.uecs.jp/uecs/kiyaku/UECSStandard100_E10.pdf
http://www.arduino.org/software
http://www.arduino.org/software

４ 機種の選定 

 この章の末尾に付録として、Arduino の主要機種のスペックを掲載しておきます。Arduino には

数多くの機種が存在しますが、大きく分けて３種類に分類されます。まず、最も標準的な

Atmega328を搭載したUNOとその性能向上型のMEGA、次にUSB通信機能をCPUに統合し、

ATmega32u4 を搭載したグループ。そして、ARM系の CPU を搭載し 3.3V で動作する新機種で

す。このうち、ARM 系機種は高性能なのですが不揮発性メモリの EEPROM を搭載していませ

ん。これは、電源を切った時に設定情報などを保持できないという大きな欠点になります。また

UARDECS で使用しているライブラリの幾つかは動作しませんので、これらの機種は対象外とし

ます。次に、Leonardをはじめとする ATmega32u4を搭載したグループですが、どの機種も UNO

よりユーザーが使用できるフラッシュメモリの総量が少ないのです。一見僅かな差ですが、

UARDECSは最低限のプログラムでも 28000byte程度を必要としますので、ATmega32u4搭載機

ではUARDECSのコンパイルはできてもフラッシュメモリ不足でプログラムを転送できないという状

況に陥ります。これは、ネットワーク通信機能を持つ Leonard Eth や Yun を使えないという点で

少々痛いのですが、筆者の PCでは Leonard と相性が悪いのか、プログラムの書き込み中にフリ

ーズし、ブートローダーを壊す事故が発生しましたので信頼性の面からもこの機種は敬遠するこ

とにしました。実は、これまで UECSの開発用に最も扱いやすいのは Arduino Ethernetだったの

ですが、非常に悔しいことにこの機種は生産終了で入手できなくなってしまいました。Arduino 

Nanoは小型機で外形が他のArduinoと異なるため、イーサネットシールドをそのままでは搭載で

きません(配線し直せば可能)。従って、現時点で使えるのは UNO と MEGAのみに絞られます。 

 

 

 

 

 

 

 

 

 

 

 

Arduino UNO(左)と Arduino MEGA 2560(右) 

５ ネットワーク通信機能の付加 

 Arduino UNO とMEGAは単体で LANポートがありません。そのため、別途 LANポートは追加

しなくてはいけません。純正品ですと Ethernet Shield 2 というものが利用でき、およそ 3000円ぐ

らいです。旧機種では Ethernet Shieldというものがありましたが、これは生産終了になっており性

能も劣るので、いまさら入手するメリットはありません。 

 

 

 

 

 

 

 

 

 

 

 

 

 

Ethernet Shield 2(左)、Ethernet Shield(右、既に生産終了) 

 

 

６ その他、必要なもの 

(1) ACアダプタ 

 Arduinoは PCとUSB接続している間は PCからの電力供給で動作しますが、PCから外して動

作させるためには別途電源が必要です。最も簡単なのは AD-L50P100 や AD-B50P200 などを

購入して PCの時と同じように USB 端子に接続することです。スマートフォン用の AC アダプタな

ども USB 端子が付いていますので代用できます。もう一つの方法は、Arduino に付いている DC

ジャックにACアダプタを接続することで、外形 5.5mm内径 2.1mmのプラグが適合します。DCジ

ャックを使用する場合、5V より少し高い電圧が必要で通常は 9V を給電します。GF12-US0913

などが使えると思われます。ただし、DC ジャックを使用する場合はレギュレータの発熱が発生し

ます。12V の AC アダプタも使用できますが、さらに発熱が多くなります。必要な電力量は LAN

接続時で最低 200mA ですので、少なくともその 2 倍は見積もったほうが良いでしょう。他に様々

なデバイスを接続すれば消費電力はその分だけ増加します。 

http://akizukidenshi.com/catalog/g/gM-08312/
http://akizukidenshi.com/catalog/g/gM-08310/
http://akizukidenshi.com/catalog/g/gM-01803/
http://akizukidenshi.com/catalog/g/gM-01803/

(2)ブレッドボードとジャンパー線 

 例えば、BB-801 や EIC-102BJ など１つぐらいはあったほうが良いです。さらに、ジャンパーケ

ーブル（オス－オス型、オス－メス型)を揃えておくと試作がやりやすくなります。 

 

(3)抵抗と LED 

 抵抗は最低限、1kΩと 10kΩがあるとプルアップやプルダウンに多用しますし、このぐらいの抵

抗をつないでおけば間違って信号線を短絡させても部品を壊さずに済みます。何をどのぐらい

揃えれば良いか分からない場合、まとめて買っても大した価格ではありません。LEDは赤色の発

光効率が良く 2V 以上の電圧で点灯できますので 1kΩの抵抗をつないで Arduino の動作確認

に使用します。 

 

(4)工具類 

 ちょっとした工作が必要になる場合は、小型のバネ付きニッパー、ワイヤーストリッパー、ハンダ

付けが必要な場合は、はんだゴテ(白光 No.984-01)、糸ハンダ(白光 FS402-01)、コテ台(コテ台

No.603)などを用意します。 

 

(5)開発用 PC とソフトウェア、LANケーブル、HUB 

 開発用 PC(WindowsPCを想定しています)にはArduino IDEをインストールします。さらに、ノー

ドの動作テスト用に UECS 用パケット送受信ツールをダウンロードして準備しておきます。複数の

ノードを連動させる実験をする場合、HUB と LANケーブルを忘れないようにしましょう。 

 

７ UARDECSのサンプルプログラム 

 UARDECS Ver0.8には以下のサンプルが付属します。 

#ApplicationGuide このアプリケーションガイドに記載されたプログラムのソースコード 

Skeleton 何もしないプログラム。コンパイルが通る最低限の宣言が記述されてい

る。 

DummyInAirTemp ユーザーがWeb インターフェースに入力した値を InAirTemp (気温)と見

なして送信させる。 

Thermostat InAirTemp (気温)を受信して、ユーザーがWebインターフェースに入力し

た値と比較し 1か 0を出力する。 

Thermostat_JP Thermostatの表示を日本語化したサンプル。 

InAirTemp_ADT7410 温度センサ ADT7410の動作テスト、実物のセンサが必要。 

SelectCCMType ユーザーが CCMの種類を選択して変更するテスト。 

InAirTemp_SHT7x 温湿度センサ SHT75/SHT71/SHT15/SHT11の動作テスト、実物のセン

サと制御用のライブラリが別途必要。 

http://akizukidenshi.com/catalog/g/gP-05294/
http://akizukidenshi.com/catalog/g/gP-00285/
http://akizukidenshi.com/catalog/g/gC-05371/
http://akizukidenshi.com/catalog/g/gP-03471/
http://akizukidenshi.com/catalog/g/gR-07820/
http://akizukidenshi.com/catalog/g/gR-07838/
http://akizukidenshi.com/catalog/g/gR-07791/
http://akizukidenshi.com/catalog/g/gI-00624/
http://www.amazon.co.jp/VESSEL-%E3%83%99%E3%83%83%E3%82%BB%E3%83%AB-No-3500E-1-%E3%83%AF%E3%82%A4%E3%83%A4%E3%83%BC%E3%82%B9%E3%83%88%E3%83%AA%E3%83%83%E3%83%91%E3%83%BC/dp/B000CED24U/ref=lp_2039375051_1_2?s=diy&ie=UTF8&qid=1457487782&sr=1-2
http://www.amazon.co.jp/%E7%99%BD%E5%85%89-%E5%8D%8A%E7%94%B0%E3%82%B4%E3%83%86-%E3%83%97%E3%83%AC%E3%82%B9%E3%83%88-%E3%82%AD%E3%83%A3%E3%83%83%E3%83%97%E4%BB%98-No-984-01/dp/B000AQSCQS
http://www.amazon.co.jp/%E7%99%BD%E5%85%89-HAKKO-No-603-%E4%B8%89%E5%BE%B3%E3%82%B3%E3%83%86%E5%8F%B0/dp/B000AQMK7K/ref=sr_1_6?s=diy&ie=UTF8&qid=1457488257&sr=1-6&keywords=%E7%99%BD%E5%85%89+%EF%BC%96%EF%BC%90%EF%BC%93
http://www.amazon.co.jp/%E7%99%BD%E5%85%89-HAKKO-No-603-%E4%B8%89%E5%BE%B3%E3%82%B3%E3%83%86%E5%8F%B0/dp/B000AQMK7K/ref=sr_1_6?s=diy&ie=UTF8&qid=1457488257&sr=1-6&keywords=%E7%99%BD%E5%85%89+%EF%BC%96%EF%BC%90%EF%BC%93
http://uecs.org/arduino/uecsrs.html

８ 何もしないプログラム 

下準備として、Ethernet Shield2を装着したArduino UNOをPCにUSB接続して下さい。Arduino 

IDE を起動し、ポートの設定画面で COMポートの設定を確認したらサンプルプログラム Skeleton

を読み込んでみます。マニュアル通りに UARDECSがインストールされていれば以下の場所から読

み込めます。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Skeletonの読み込み 

以下に Skeletonのソースコードを示します（簡略化のためコメント行を消しました）。読み込んだら

IDE 左上の→ボタンを押して Arduino に書き込んでみてください。このプログラムは、単にコンパイ

ルを通るだけで、CCM が１つも作成されていません。そのため、まだ UECS ノードとしては機能しま

せん。ソースコード内に書かれているのは、UARDECS の起動に最低限必要な設定項目で、セー

フモードジャンパーの設定(IP アドレスが分からなくなった場合に強制的に IP アドレスを設定するス

イッチ)、ノードの機種名、製造元、UECSID、ノードの初期名、MACアドレスなどです。（注意: MAC

アドレスはノード毎に違う値が必要です、Ethernet Shield2の裏に書いてある値を入力して下さい） 

 

//Sample skeleton 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

 

const int U_MAX_CCM = 0; 

UECSCCM U_ccmList[U_MAX_CCM]; 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

 

//本当はここで CCMを初期化する 

 

} 

void OnWebFormRecieved(){} 

void UserEverySecond(){} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

 

void loop() 

{ 

UECSloop(); 

} 

 

void setup() 

{ 

UECSsetup(); 

} 

 

Skeletonの状態でもWebサーバは機能しています。LANケーブルで Arduino を PCに接続し、

ブラウザでアクセスすると以下のような画面を見ることができます。(工場出荷時の Arduinoでは IP

アドレスが 192.168.1.7になります。詳細は UARDECSマニュアル５章を参照) 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Skeletonを実行中の Arduinoにアクセスした場合の応答 

９ CCMを作成して自動送信させる 

Skeletonに CCMを１つ作成するには以下の場所を変更します。この例では” cnd.xXX”という送

信 CCMを作成しました。青字が変更箇所、赤字が新しく追加された箇所です。１つの CCMに３種

類の文字列(CCMの説明,type,単位)が必要です。この文字列は必ず関数の外で宣言してグロー

バル変数にします。次に UserInit()の中で UECSsetCCM関数を実行してCCMを宣言します。この

サンプルプログラムは#ApplicationGuideの中の Sample1SendCCMに収録されています。 

 

//Sample1SendCCM 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

 

const int U_MAX_CCM = 1;//CCMの総数を 1に 

UECSCCM U_ccmList[U_MAX_CCM]; 

//CCM定義用の素材、被らないように適当な変数名で 3つ宣言(必ず PROGMEMを付ける) 

const char ccmInfoTest[] PROGMEM= "Test";//CCMの説明(Webでのみ表示) 

const char ccmTypeTest[] PROGMEM= "cnd.xXX";//CCMの Type文字列 

const char ccmUnitTest[] PROGMEM= "";//CCMの単位(この場合単位無し) 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

//UECSsetCCM(送受信の区分[trueで送信],通し番号[0から始まる],CCM説明,Type,単

位,priority[通常は 29],少数桁数[0で整数],送信頻度設定[A_1S_0 で 1秒間隔]) 

UECSsetCCM(true,0 ,ccmInfoTest,ccmTypeTest,ccmUnitTest,29,0,A_1S_0); 

} 

void OnWebFormRecieved(){} 

void UserEverySecond(){} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

 

void loop() 

{ 

UECSloop(); 

} 

 

void setup() 

{ 

UECSsetup(); 

} 

PCに LANケーブルで繋いだ Arduinoにこのプログラムを書き込んでから UECS用パケット送受

信ツールで見てみると、1秒間隔で CCMが送信されているのが分かります。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

CCM作成後の応答を UECS用パケット送受信ツールで見た時の画面

http://uecs.org/arduino/uecsrs.html

１０ アナログ入力値の送信 

先ほどのプログラムでは、作成した送信CCMが自動送信される様子を示しましたが、そのままで

は値を設定していないので常に 0が出力されるだけです。そこで、Arduinoのアナログ入力値を

CCM として出力してみます。変更点は以下の色付きの箇所です。U_ccmList[0]の[0]の部分は

UECSsetCCM関数の 2番目の引数で指定した通し番号です。このサンプルプログラムは

#ApplicationGuideの中の Sample2SendAnalogValに収録されています。 

 

 

 

//Sample2SendAnalogVal 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

 

const int U_MAX_CCM = 1;//CCMの総数を 1に 

UECSCCM U_ccmList[U_MAX_CCM]; 

//CCM定義用の素材、被らないように適当な変数名で 3つ宣言(必ず PROGMEMを付ける) 

const char ccmInfoTest[] PROGMEM= "Test";//CCMの説明(Webでのみ表示) 

const char ccmTypeTest[] PROGMEM= "cnd.xXX";//CCM の Type文字列 

const char ccmUnitTest[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

//UECSsetCCM(送受信の区分[trueで送信],通し番号[0から始まる],CCM 説明,Type,単

位,priority[通常は 29],少数桁数[0で整数],送信頻度設定[A_1S_0 で 1秒間隔]) 

UECSsetCCM(true,0 ,ccmInfoTest,ccmTypeTest,ccmUnitTest,29,0,A_1S_0); 

} 

void OnWebFormRecieved(){} 

 

void UserEverySecond() 

{ 

U_ccmList[0].value=analogRead(0);//CCM に ADコンバータの値を書き込む 

} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

 

void loop() 

{ 

UECSloop(); 

} 

 

void setup() 

{ 

UECSsetup(); 

} 

 

プログラムではArduinoのA0端子の電圧を読み取るように設定しました。次に、UNOのA0と 5V

を 1kΩの抵抗でつないでみます。実は抵抗無しで直結しても大丈夫ですが、安全のためです。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

結線箇所（5V） 

※図は UNO本体ですが、Ethernet Shield 2を載せていると見なして下さい 

5V入力時の応答 

A0端子に 5Vを入力した結果、CCMの値は 1023に変化しました。Arduino UNO と MEGAの

AD コンバータは標準設定で 0～5Vの電圧を 0～1023で示すので、これで正常です。次に、A0を

3.3Vの端子に接続してみます。 

 

 

 

 

 

 

 

 

 

 

 

 

 

結線箇所(3.3V) 

3.3V入力時の応答 

 

A0端子に 3.3Vを入力した結果 CCMの値は大体 694～695の間を行ったり来たりしているのが

分かります(もしかすると個体差があるかもしれません)。ここから実際の電圧を計算する場合、694.5

×5÷1023≒3.4V となり、大体正しい電圧を示していることが分かります。これで電圧を CCM とし

て出力するノードが作れます。 

注意点ですが、A0端子に何も接続しない場合、空気中の静電気を拾ってしまい、出てくる値は

メチャメチャになります。これはピンをデジタル入力として使う場合も同じで、何も接続されていない

ピンの値は不定になります。 

 

１１ セーフモードについて 

 サンプルプログラムではD3ピンをGNDに繋がない限り常にセーフモードで起動するようになって

います。セーフモードではノードの設定値が初期化され、IPアドレスも自由に設定できません。ノー

ドの IPアドレスなどを自由に設定したい場合、D3ピンを 1kΩの抵抗を介してGNDに接続するか、 

 const byte U_InitPin_Sense=HIGH; 

と記述された行を 

 const byte U_InitPin_Sense=LOW; 

に変更してセーフモードを抜けるようにして下さい。 

付録  Arduino の機種 

 Arduino には大変多くの機種があり、一見するとどれがどう違うのか分からなくなりますが、大雑

把に特徴をまとめておきます。ただし、ここに表記されているのは出回っている一部のみであり、

互換機を含めればさらに膨大な種類があります。 

 

機種名（一部） 
駆動電圧 

(V) 
搭載ポート 

全入出力ピン 

(アナログ入力/出力) 
メイン CPU サブデバイス 

Arduino UNO R3 5 USB 20(6) ATmega328 
ATmega16u2 

(USB-シリアル変換) 

Arduino Mega 2560 R3 5 USB 70(16) ATmega2560 
ATmega16u2 

(USB-シリアル変換) 

Arduino Ethernet R3 ※1 5 LAN 20(6) ATmega328 
W5100 

(LAN 制御用) 

Arduino Nano Ver3.x 5 MiniUSB 22(8) ATmega328 
FT232RL 

(USB-シリアル変換) 

Arduino Leonardo 5 MicroUSB 20(12)※2 ATmega32u4   

Arduino Leonardo Eth 5 
MicroUSB 

LAN 
20(12)※2 ATmega32u4 

W5500 

(LAN 制御用) 

Arduino Micro 5 MicroUSB 20(14) ATmega32u4   

Arduino Yun 5 

MicroUSB 

USB 

LAN Wifi 

20(12)※2 ATmega32u4 
Atheros AR9331 

(LAN/Wifi 制御用) 

Arduino Due 3.3 MicroUSBx2 70(12/2)※3 SAM3X8E 
ATmega16u2 

(USB-シリアル変換) 

Arduino M0 3.3 MicroUSB 20(6/1)※3 SAMD21G18A   

Arduino M0 Pro 3.3 MicroUSBx2 20(6/1)※3 SAMD21G18A 
Atmel a06-0736 

(オンボードデバッガ) 

※1 既に生産終了 

※2 ICSP端子も使用する場合、総ピン数+3 

※3 アナログ入力端子はデジタル入出力には使えない、デジタル入力端子に内蔵プルアップ

機能が無い 

機種名（一部） 
フラッシュメモリ 

(byte) 

SRAM 

(kbyte) 

EEPROM 

(kbyte) 
特徴 

Arduino UNO R3 32256 2 1 
最も基本的な機種。 

単に Arduino と言ったらこれを示す。 

Arduino Mega 2560 R3 253952 8 4 

UNO のピン数とメモリを大幅に増量したも

の。処理速度は変わらない。一部のピンの機

能は UNO と異なるので注意が必要。 

Arduino Ethernet R3 32256 2 1 
UNOから USBを外し Ethernet Shieldの機能

を付加したもの。 

Arduino Nano Ver3.x 30720 2 1 
UNO の小型版。形状が全く異なる。UNO より

ピン数が増えている。 

Arduino Leonardo 28672 2.5 1 

CPU に USB-シリアル変換機能を統合して回

路を単純化したものだが、内部配線が UNO

とはかなり異なる。UNO よりプログラム用メモ

リが少ない。PC に接続し擬似的なキーボー

ドやマウスのように動作させる事ができる。 

Arduino Leonardo Eth 28672 2.5 1 
Leonardo に Ethernet Shield 2 の機能を付加

したもの。 

Arduino Micro 28672 2.5 1 
Leonardoの小型版だが Nanoともピン配置が

異なる。 

Arduino Yun 28672 2.5 1 
LeonardoにWifi制御用のCPUを別途搭載し

たもの。 

Arduino Due 524288 96 無し 

外観は Mega に似ているが搭載 CPU が異な

るため、全く別物である。駆動電圧の違いに

注意。EEPROM を搭載しない。 

Arduino M0 262144 32 無し 
高性能な次世代機。CPUが違うため UNO と

は全く別物。EEPROM を搭載しない。 

Arduino M0 Pro 262144 32 無し 
M0 にデバッガを搭載した開発用。EEPROM

を搭載しない。 

 

 

（西日本農業研究センター 黒崎秀仁） 

第２章 日射量を計測する 

 

１ はじめに 

 温室向けの日射量の測定は遮光カーテンの制御などに必要ですが、昔からある日射センサは精

度が高いものの１個１０万円ぐらいする上に、出力がミリボルト単位と微小で、計測にも高価なデー

タロガーが必要でなかなか農業用には向かないものでした。しかし、最近 1万 5000円ぐらいで買え

る安価な日射センサが登場しました。PVアレイ日射計です。 

 

 

 

 

 

 

 

 

 

 

 

 

PVアレイ日射計(三弘) 

 

 PVアレイ日射計の出力は 1.0kW/m2あたり 1.0V と分かりやすい値で、日射センサの中では出力

が大きく、Arduinoのアナログ入力端子に接続して測定できるだけの十分な電圧があります。 

 

２ ソフトウェアと結線 

 前章をご覧になった方はすぐに分かると思いますが、基本部分は特集１で示したサンプルプログ

ラムを少し書き換えるだけですぐに作れてしまいます。今回は CCMを 2つ作成し、片方は

"cnd.mOC"としました。".mOC"の部分は屋外気象観測装置を示す種別名で、この部分の付け方

は通信実用規約 1.00-E10の 31ページを参照して下さい。もう片方は"WRadiation"ですが、これは

屋外日射を示します。こちらは予約CCMに含まれている（通信規約 41ページ）ので種別名を省略

しています。 

 次に、アナログ入力値を日射量への換算する方法ですが、1.0Vが入力された時の AD変換値は

1023÷5 = 204.6 となりますので、AD変換値 × (1／204.6) = 日射量(kW/m2) となります。ただ

し、整数値で演算すると小数点以下が切り捨てられてしまうので、サンプルではあえて分かりやす

いように一度少数が扱える変数を介して計算させました。次に、小数の表現ですが、Arduinoの仕

http://www.sanko-web.co.jp/assist2/20140604_PVSS-01.pdf
http://www.uecs.jp/uecs/kiyaku/UECSStandard100_E10.pdf

様では floatや double型の変数は浮動小数点数を扱えますが、UARDECS で CCMに出力すると

きは一度整数に直す必要があります。少々ややこしいですが、CCMの値の小数点の位置は

UECSsetCCM()関数で宣言した時の 7番目の引数の値で決まります。例えば、以下のように少数

桁数に 3を指定した場合、 

 

 UECSsetCCM(true,1,ccmInfoRad,ccmTypeRad,ccmUnitRad,29,3,A_1S_0); 

 

 U_ccmList[1].valueの値が 1234の時、実際に CCM として出力される値は 1.234になります。少

数桁数が 3の時に double型の変数からU_ccmList[1].valueに値を代入する場合、桁を合わせるた

めに 1000倍してから代入しなければなりません(少数桁数が 1の時は 10倍、2の時は 100倍して

下さい)。これらの処理を施したのが以下のプログラムです。このサンプルプログラムは

#ApplicationGuideの中の Sample3SendRadiationに収録されています。 

 

 

//Sample3SendRadiation 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

 

const int U_MAX_CCM = 2;//CCMの総数を 2に 

UECSCCM U_ccmList[U_MAX_CCM]; 

//CCM定義用の素材、被らないように適当な変数名で 3つ宣言(必ず PROGMEMを付ける) 

const char ccmInfoTest[] PROGMEM= "NodeCondition";//CCMの説明(Webでのみ表示) 

const char ccmTypeTest[] PROGMEM= "cnd.mOC";//CCMの Type文字列 

const char ccmUnitTest[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

const char ccmInfoRad[] PROGMEM= "Radiation";//CCMの説明(Web でのみ表示) 

const char ccmTypeRad[] PROGMEM= "WRadiation";//CCMの Type文字列(屋外日射) 

const char ccmUnitRad[] PROGMEM= "kW m-2";//CCMの単位 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

 

//UECSsetCCM(送受信の区分[trueで送信],通し番号[0から始まる],CCM 説明,Type,単

位,priority[通常は 29],少数桁数,送信頻度設定[A_1S_0で 1秒間隔]) 

UECSsetCCM(true,0,ccmInfoTest,ccmTypeTest,ccmUnitTest,29,0,A_1S_0); 

UECSsetCCM(true,1,ccmInfoRad,ccmTypeRad,ccmUnitRad,29,3,A_1S_0); 

} 

void OnWebFormRecieved(){} 

 

void UserEverySecond() 

{ 

double radiation= (double)analogRead(0)*1.0/204.6; 

U_ccmList[1].value= radiation*1000; //少数桁が 3なので 1000倍 

} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

 

void loop() 

{ 

UECSloop(); 

} 

 

void setup() 

{ 

UECSsetup(); 

} 

 PVアレイ日射計を使う場合、ケーブルの赤、白はまとめて A0に、黒は GNDに接続します。これ

は 1個だけ接続する場合の配線で、複数並列して使う方法もありますが、使い方はセンサのマニュ

アルに書いてあります。ケーブルが太くて直接Arduinoへの接続が難しい場合はジャンパーケーブ

ルをハンダ付けしてしまいましょう。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Arduinoへの接続(赤白を A0に、黒を GNDに) 

※図は UNO本体ですが、Ethernet Shield 2 を載せていると見なして下さい 

３ 出力値を見てみる  

 
出力値の例 

 

 WRadiationの部分に日射量が出力されます。日射計にはかなり強い光を当てないと大きな値は

出力されません。一番日射の強い時間帯でも 1.2 kW/m2行くかどうかといったレベルです。一応、

これでも日射計としては成立するのですが、残念なところは0～5Vの範囲を1023段階で測れるAD

コンバータで 0～1Vぐらいを行ったり来たりする値を測っているので本来の 1/5の範囲しか活用で

きていないことになります。そのため、分解能が低くなってしまい、出力値だけは小数点以下 3桁も

あるのに、実際には 200段階ぐらいの分解能しかありません。次はこの解決策を解説します。 

４ AD コンバータの計測レンジを変更する 

 実は、Arduinoにはソフトウェア上で AD コンバータの計測レンジを変更できる機能があります。

analogReference()関数による基準電圧の変更です。設定できる値は機種により異なり、UNOでは

5V、1.1Vの 2種類、MEGAでは 5V、2.56V、1.1Vの 3種類から選ぶことができます。機種により記

号定数が違うので注意して下さい。 

 

UNOの場合、 

 analogReference(DEFAULT); //UNO基準電圧 5V 

 analogReference(INTERNAL); //UNO基準電圧 1.1V 

 

MEGAの場合 

 analogReference(DEFAULT); //MEGA基準電圧 5V 

 analogReference(INTERNAL1V1); //MEGA基準電圧 1.1V 

 analogReference(INTERNAL2V56); //MEGA基準電圧 2.56V 

 

 例えば、1.1Vを指定すると 0～1.1Vの範囲を 1023段階で表すようになります。当然ながら、日射

量への換算式も変わってきます。基準電圧を 1.1Vに設定した場合の 1.0Vが入力された時の AD

変換値は 1023÷1.1 = 930 となりますので、AD変換値 × (1／930) = 日射量(kW/m2)となります。

これらの処理を入れたものが以下のプログラムです。このサンプルプログラムは#ApplicationGuide

の中の Sample4ArefChangeに収録されています。 

 

//Sample4ArefChange 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

const int U_MAX_CCM = 2;//CCMの総数を 2に 

UECSCCM U_ccmList[U_MAX_CCM]; 

//CCM定義用の素材、被らないように適当な変数名で 3つ宣言(必ず PROGMEMを付ける) 

const char ccmInfoTest[] PROGMEM= "NodeCondition";//CCMの説明(Webでのみ表示) 

const char ccmTypeTest[] PROGMEM= "cnd.mOC";//CCMの Type文字列 

const char ccmUnitTest[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

const char ccmInfoRad[] PROGMEM= "Radiation";//CCMの説明(Web でのみ表示) 

const char ccmTypeRad[] PROGMEM= "WRadiation";//CCMの Type文字列(屋外日射) 

const char ccmUnitRad[] PROGMEM= "kW m-2";//CCMの単位 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

 

//UECSsetCCM(送受信の区分[trueで送信],通し番号[0から始まる],CCM 説明,Type,単

位,priority[通常は 29],少数桁数,送信頻度設定[A_1S_0で 1秒間隔]) 

UECSsetCCM(true,0,ccmInfoTest,ccmTypeTest,ccmUnitTest,29,0,A_1S_0); 

UECSsetCCM(true,1,ccmInfoRad,ccmTypeRad,ccmUnitRad,29,3,A_1S_0); 

} 

void OnWebFormRecieved(){} 

 

void UserEverySecond() 

{ 

double radiation= (double)analogRead(0)*1.0/930.0; 

U_ccmList[1].value= radiation*1000; //少数桁が 3なので 1000倍 

} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

 

void loop() 

{ 

UECSloop(); 

} 

 

void setup() 

{ 

analogReference(INTERNAL);//UNO基準電圧 1.1V 

UECSsetup(); 

}  

５ AD コンバータの計測レンジの自動変更 

 日射量はほとんどの時間帯で 1.1kW/m2以下に収まるのですが、厄介なことに日射量の多いごく

僅かな時間帯だけ 1.2 kW/m2ぐらいになり、計測レンジを 1.1Vに固定すると微妙に上限を超えて

しまいます。上限を超えると analogRead(0)の戻り値が 1023になりますので、一度測ってみて超えて

いそうだったら計測レンジを 5Vに戻して再計測することで解決できます。ただし、こちらで実験した

ところ、基準電圧を変更してから少しの間は正常な測定ができません。基準電圧の変更に 10 ミリ秒

を要し、そこから安定した計測ができるまで数回 analogRead()を実行して読み飛ばす必要がありま

す。これらの処理を実装するとこうなります。cnd.mOCには計測レンジの状態を出力するようにしま

した。このサンプルプログラムは#ApplicationGuideの中の Sample5AutoRangeに収録されていま

す。 

 

//Sample5AutoRange 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

 

const int U_MAX_CCM = 2;//CCMの総数を 2に 

UECSCCM U_ccmList[U_MAX_CCM]; 

//CCM定義用の素材、被らないように適当な変数名で 3つ宣言(必ず PROGMEMを付ける) 

const char ccmInfoTest[] PROGMEM= "NodeCondition";//CCMの説明(Webでのみ表示) 

const char ccmTypeTest[] PROGMEM= "cnd.mOC";//CCMの Type文字列 

const char ccmUnitTest[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

const char ccmInfoRad[] PROGMEM= "Radiation";//CCMの説明(Web でのみ表示) 

const char ccmTypeRad[] PROGMEM= "WRadiation";//CCMの Type文字列(屋外日射) 

const char ccmUnitRad[] PROGMEM= "kW m-2";//CCMの単位 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

 

//UECSsetCCM(送受信の区分[trueで送信],通し番号[0から始まる],CCM 説明,Type,単

位,priority[通常は 29],少数桁数,送信頻度設定[A_1S_0で 1秒間隔]) 

UECSsetCCM(true,0,ccmInfoTest,ccmTypeTest,ccmUnitTest,29,0,A_1S_0); 

UECSsetCCM(true,1,ccmInfoRad,ccmTypeRad,ccmUnitRad,29,3,A_1S_0); 

} 

 

void OnWebFormRecieved(){} 

 

void UserEverySecond() 

{ 

static bool lowRange=false; 

double radiation=analogRead(0); 

if(radiation==1023 && lowRange) 

  { 

//1.1Vモードでレンジ外になった場合、5Vモードに変更 

    lowRange=false; 

    analogReference(DEFAULT);//UNO基準電圧 5V 

    analogRead(0);//読み飛ばす 

    return; 

  } 

 

if(radiation<200 && !lowRange) 

  { 

//5Vモードで 1V未満になった場合、1.1Vモードに変更 

    lowRange=true; 

    analogReference(INTERNAL);//UNO基準電圧 1.1V 

    analogRead(0);//読み飛ばす 

    return; 

  } 

 

if(lowRange) 

  { 

  //1.1Vモード換算式 

  radiation= radiation*1.0/930.0; 

  } 

 else 

  { 

  //5Vモード換算式 

  radiation= radiation*1.0/204.6; 

  } 

  U_ccmList[1].value= radiation*1000; //少数桁が 3なので 1000 倍 

  U_ccmList[0].value= lowRange;//現在の計測レンジを出力 

} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

 

void loop() 

{ 

UECSloop(); 

} 

 

void setup() 

{ 

analogReference(DEFAULT);//UNO基準電圧 5V 

UECSsetup(); 

} 

  

 

 実際には 5Vモードの作動頻度はごく僅かだと思われます。この方式の難点は、レンジを切り替え

る時に１秒間計測ができずに過去の値が出力されてしまうところで、急激に値が変化し続ける時に

は追従できません。また、5Vの電源電圧が正確に供給されていないと 5Vモードで出力値が狂う可

能性があります。もし、MEGAを使う場合、妥協して最初から基準電圧を 2.56Vモードで固定してし

まえば、まあまあの精度で計測でき、通常の日射量の範囲であればレンジを超えることは無いと思

われます。なお、実際の温室で使用する場合は十分な動作検証を行い、自己責任の上で使用し

て下さい。 

 

（西日本農業研究センター 黒崎秀仁） 

第３章 LEDの点灯・消灯 

 

１ 受信 CCMの作成 

 これまでは CCMの送信ノードについて解説してきましたが、次は特定の CCMを受信することで

何らかの動作を起こすノードの作り方を解説します。そのためには受信 CCMの扱いについて説明

する必要があります。 

 受信CCMを作る場合も送信CCMと同じように３種類の文字列(CCMの説明,type,単位)を作成し

ます。次にUECSsetCCM関数を記述するとき、最初の引数を falseにすると受信CCMになります。

例えば以下のように記述します。 

 

 UECSsetCCM(false,1,ccmInfoRcv,ccmTypeRcv,ccmUnitRcv,29,0,A_1S_0); 

 

 今度も Skeletonから改造してみます。以下のプログラムでは送信 CCMを１個、受信 CCMを１個

作りました。改造箇所は色付きの部分です。このサンプルプログラムは#ApplicationGuideの中の

Sample6ReceiveCCMに収録されています。 

 

 

//Sample6ReceiveCCM 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

const int U_MAX_CCM = 2;//CCMの総数を 2に 

UECSCCM U_ccmList[U_MAX_CCM]; 

//CCM定義用の素材 1、被らないように適当な変数名で 3つ宣言 

const char ccmInfoCnd[] PROGMEM= "NodeCondition";//CCMの説明(Webでのみ表示) 

const char ccmTypeCnd[] PROGMEM= "cnd.xXX";//CCMの Type文字列 

const char ccmUnitCnd[] PROGMEM= "";//CCM の単位(この場合単位無し) 

//CCM定義用の素材 2、被らないように適当な変数名で 3つ宣言 

const char ccmInfoRcv[] PROGMEM= "Receive CCM";//CCMの説明(Webでのみ表示) 

const char ccmTypeRcv[] PROGMEM= "Receive.xXX";//CCMの Type文字列 

const char ccmUnitRcv[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

 

//UECSsetCCM(送受信の区分,通し番号,CCM 説明,Type,単位,priority,少数桁数,送信頻度

設定[A_1S_0で 1秒間隔]) 

UECSsetCCM(true,0,ccmInfoCnd,ccmTypeCnd,ccmUnitCnd,29,0,A_1S_0);//送信 CCM 

UECSsetCCM(false,1,ccmInfoRcv,ccmTypeRcv,ccmUnitRcv,29,0,S_1S_0);//受信 CCM 

} 

void OnWebFormRecieved(){} 

void UserEverySecond(){} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

void loop() 

{ 

UECSloop(); 

} 

void setup() 

{ 

UECSsetup(); 

} 

Web インターフェースの状態 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

CCMの送信状態 

 

 UNOに上記プログラムを転送してブラウザでアクセスすると、”cnd.xXX”と”Recieve.xXX”の 2種

類の CCMが確認できます。UDPパケットの方は”cnd.xXX”が１秒間隔で送信されているのが分か

ります。しかし、まだこの段階では CCMを受信させても何も起こりません。 

２ 受信 CCMの値へのアクセス 

 次の改造ですが、受信CCMの値をそのまま送信CCMに代入してみます。ただし、受信CCMは

常に有効な値が受信できているかどうか確認してから処理する必要があります。値の有効/無効は

U_ccmList[通し番号].validity という変数を参照することで判定できます。このサンプルプログラムは

#ApplicationGuideの中の Sample7EchoCCMに収録されています。 

//Sample7EchoCCM 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

UECSOriginalAttribute U_orgAttribute; 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

 

const int U_MAX_CCM = 2;//CCMの総数を 2に 

UECSCCM U_ccmList[U_MAX_CCM]; 

//CCM定義用の素材 1、被らないように適当な変数名で 3つ宣言 

const char ccmInfoCnd[] PROGMEM= "NodeCondition";//CCMの説明(Webでのみ表示) 

const char ccmTypeCnd[] PROGMEM= "cnd.xXX";//CCMの Type文字列 

const char ccmUnitCnd[] PROGMEM= "";//CCM の単位(この場合単位無し) 

//CCM定義用の素材 2、被らないように適当な変数名で 3つ宣言 

const char ccmInfoRcv[] PROGMEM= "Receive CCM";//CCMの説明(Webでのみ表示) 

const char ccmTypeRcv[] PROGMEM= "Receive.xXX";//CCMの Type文字列 

const char ccmUnitRcv[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

 

//UECSsetCCM(送受信の区分,通し番号,CCM 説明,Type,単位,priority,少数桁数,送信頻度

設定[A_1S_0で 1秒間隔]) 

//送信 CCM 

UECSsetCCM(true,0,ccmInfoCnd,ccmTypeCnd,ccmUnitCnd,29,0,A_1S_0); 

//受信 CCM 

UECSsetCCM(false,1,ccmInfoRcv,ccmTypeRcv,ccmUnitRcv,29,0,S_1S_0); 

} 

void OnWebFormRecieved(){} 

void UserEverySecond() 

{ 

//受信 CCMの値を送信 CCMに代入 

if(U_ccmList[1].validity) 

  { 

  U_ccmList[0].value=U_ccmList[1].value;//値が有効な場合 

  } 

else 

  { 

  U_ccmList[0].value=0;//値が無効の場合 0にする 

  } 

} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

void loop() 

{ 

UECSloop(); 

} 

void setup() 

{ 

UECSsetup(); 

} 

CCMの送信テスト 

 CCMの送信テストには PC と Arduinoを LANケーブルで直結し、UECS用パケット送受信ツール

を使用します。①の欄の type=のところを“ Receive.xXX”に書き換え、値の部分を適当な数値に変

更、②の部分で IP指定を選び送信先 Arduinoの IPアドレスを入力、③の１回送信を押します。正

常に受信されれば、④の“cnd.xXX”の値が①で入力した値に 3秒間だけ変化し、その後また 0に

戻ります。これで、送った CCMに対して Arduinoが応答していることが分かりました。 

 1回の送信に対して 3秒間だけ値が帰ってくる理由は、UECSsetCCM()関数の最後の引数を

S_1S_0にしたのが原因です。UECSの通信実用規約 1.00-E10では送受信頻度ごとに、CCMの有

効期限が定められており、一般に送信間隔の3倍が有効期限です。したがって、1秒間隔で送信さ

れる S_1S_0の有効期限は受信してから 3秒間です。これを 1分間隔で送信される S_1M_0に書き

換えた場合、値は 3分間有効と見なされます。 

 

※A_1S_0 と S_1S_0の違いは、A_??_??の CCMは常時一定時間おきに送信されるのに対して

S_??_??の CCMは必要なときだけ一定時間おきに送信され、不要なときは停止することを示してい

ます。 

http://www.uecs.jp/uecs/kiyaku/UECSStandard100_E10.pdf

３ LEDの点灯 

 次に 0以外の値を受信した時、LEDを点灯するように改造してみます。LEDは D2ピンに接続す

ることにしました。どのピンが利用できるかは UARDECSマニュアルの３章を見て下さい。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

LEDの結線（LEDは足の長いほうを D2側、短い方を GND側に接続） 

※図は UNO本体ですが、Ethernet Shield 2 を載せていると見なして下さい 

 

 

 次にサンプルプログラムを示します。修正点として、setup()関数内でピンの初期化を行っていま

す。先ほどの”cnd.xXX”の使い方は、微妙に UECSの通信実用規約 1.00-E10から外れていたの

で、”cnd.xXX”には LEDが点灯している間 1を出力し、それ以外は 0になるようにしました。”cnd”

が付いた CCMにはノードの動作状態を示す値を出力することになっており、整数値で 0～4095の

範囲はユーザーが自由に意味を定義して使うことができます。このサンプルプログラムは

#ApplicationGuideの中の Sample8LEDTestに収録されています。 

http://www.uecs.jp/uecs/kiyaku/UECSStandard100_E10.pdf

//Sample8LEDTest 

#include <SPI.h> 

#include <Ethernet2.h> //Arduino IDE Ver1.7.2以降で W5500搭載機種 

//#include <Ethernet.h> //Ver1.7.2以降で W5100搭載機種 

#include <avr/pgmspace.h> 

#include <EEPROM.h> 

#include <Uardecs.h> 

 

const byte U_InitPin = 3;//このピンは変更可能です 

const byte U_InitPin_Sense=HIGH; 

const char U_name[] PROGMEM= "UARDECS Node v.1.0"; 

const char U_vender[] PROGMEM= "XXXXXX Co."; 

const char U_uecsid[] PROGMEM= "000000000000"; 

const char U_footnote[] PROGMEM= "Test node"; 

char U_nodename[20] = "Sample"; 

 

UECSOriginalAttribute U_orgAttribute; 

const int U_HtmlLine = 0; 

struct UECSUserHtml U_html[U_HtmlLine]={}; 

const int U_MAX_CCM = 2;//CCMの総数を 2に 

UECSCCM U_ccmList[U_MAX_CCM]; 

 

//CCM定義用の素材 1、被らないように適当な変数名で 3つ宣言 

const char ccmInfoCnd[] PROGMEM= "NodeCondition";//CCMの説明(Webでのみ表示) 

const char ccmTypeCnd[] PROGMEM= "cnd.xXX";//CCMの Type文字列 

const char ccmUnitCnd[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

//CCM定義用の素材 2、被らないように適当な変数名で 3つ宣言 

const char ccmInfoRcv[] PROGMEM= "Receive CCM";//CCMの説明(Webでのみ表示) 

const char ccmTypeRcv[] PROGMEM= "Receive.xXX";//CCMの Type文字列 

const char ccmUnitRcv[] PROGMEM= "";//CCM の単位(この場合単位無し) 

 

void UserInit(){ 

U_orgAttribute.mac[0] = 0x00; 

U_orgAttribute.mac[1] = 0x00; 

U_orgAttribute.mac[2] = 0x00; 

U_orgAttribute.mac[3] = 0x00; 

U_orgAttribute.mac[4] = 0x00; 

U_orgAttribute.mac[5] = 0x00; 

//UECSsetCCM(送受信の区分,通し番号,CCM 説明,Type,単位,priority,少数桁数,送信頻度

設定[A_1S_0で 1秒間隔]) 

//送信 CCM 

UECSsetCCM(true,0,ccmInfoCnd,ccmTypeCnd,ccmUnitCnd,29,0,A_1S_0); 

//受信 CCM 

UECSsetCCM(false,1,ccmInfoRcv,ccmTypeRcv,ccmUnitRcv,29,0,S_1S_0); 

} 

void OnWebFormRecieved(){} 

void UserEverySecond() 

{ 

//受信 CCMが有効かつ 0以外 

if(U_ccmList[1].validity && U_ccmList[1].value) 

  { 

  U_ccmList[0].value=1; 

  digitalWrite(2,HIGH); 

  } 

else 

  { 

  U_ccmList[0].value=0; 

  digitalWrite(2,LOW); 

  } 

} 

void UserEveryMinute(){} 

void UserEveryLoop(){} 

void loop() 

{ 

UECSloop(); 

} 

void setup() 

{ 

pinMode(2,OUTPUT); 

UECSsetup(); 

} 

CCMの送信テスト２ 

 先ほどの実験同様UECS用パケット送受信ツールを使用し、PCとArduinoを繋いでテストします。

①の欄の type=のところを“ Receive.xXX”に書き換え、値を 0以外に変更、②の部分で IP指定を

選び Arduinoの IPアドレスを入力、③の１回送信を押します。正常に受信されれば、④の

“cnd.xXX”の値が３秒間だけ１に変化し、その間、下図のようにLEDが点灯します。また、⑤の自動

送信ボタンを押せばずっと LEDを点灯させ続けることができます。 

 

 

 

 

 

 

 

 

 

 

 

LEDの点灯 



37 

 

 前述したプログラムでLEDを点灯させ続けるにはCCMを延々と送信し続ける必要がありましたが、

UserEverySecond()関数を次のように書き換えれば、“Receive.xXX”に 0以外の値を一度送信すれ

ばずっと点灯を続け、消したいときは 0を送信して消灯できます。ただし、CCMは通信環境が悪い

とパケットロスが発生し、届かないことがあります。この例はパケットロスが起きると送信された CCM

と動作が一致しなくなるので、パケットロスがあっても問題にならないような場面に使用することをお

勧めします。このサンプルプログラムは#ApplicationGuideの中の Sample8LEDTestLatchに収録さ

れています。 

 

//ラッチ式点灯・消灯 

void UserEverySecond() 

{ 

if(U_ccmList[1].validity && U_ccmList[1].value)//点灯 

  { 

  U_ccmList[0].value=1; 

  digitalWrite(2,HIGH); 

  } 

if(U_ccmList[1].validity && U_ccmList[1].value==0)//消灯 

  { 

  U_ccmList[0].value=0; 

  digitalWrite(2,LOW); 

  } 

} 

 一方、点灯のCCMを送った後、タイムアウトするまでの時間を３秒より長く設定したい場合、以下

のように書き換えて、自前で点灯時間を管理できます。このサンプルプログラムは

#ApplicationGuideの中の Sample8LEDTestTimeout10sに収録されています。 

 

//タイムアウト時間を 10秒にする 

void UserEverySecond() 

{ 

if(U_ccmList[1].validity && U_ccmList[1].value) 

  { 

  U_ccmList[0].value=1; 

  digitalWrite(2,HIGH); 

  } 

else if(U_ccmList[1].flagStimeRfirst && U_ccmList[1].recmillis>10000)//10sec 

  { 

  U_ccmList[0].value=0; 

  digitalWrite(2,LOW); 

  } 

} 

 

 U_ccmList[1].recmillisには、最後に CCMを受信してからの経過時間がミリ秒単位で記録されて

おり、細かく CCMの受信タイミングを知ることができます。この値は最大 24時間分までカウントアッ

プされます(誤差あり)。ただし、U_ccmList[1].flagStimeRfirstが trueでない場合、

U_ccmList[1].recmillisの値の正確さは保証されませんので、必ず確認してから参照します。 

４ LEDを別のデバイスに置き換える 

 これまでの説明で、CCM に応答する最低限の UECS ノードのプロトタイプができたことになります。

LEDの点灯をコントロールできるということは LEDの代わりに様々なものを接続してスイッチの

ON/OFFを制御できるということです。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

9V リレーの動作回路例 

 ここに LEDの代わりにリレーを接続した例を示します。ここでは 941H-2C-9Dを使用しま

した。このリレーは 9Vで動作するので DCジャックに 9Vの ACアダプタを接続する必要が

あります。リレーの電源は DCジャックの＋側に直結された Vin端子から取っています。9V

の電圧は直接 Arduinoでは扱えませんので、間にトランジスタ 2SC1815を入れています。

様々な文献で言われていますが、リレーは作動するときにサージ対策が必要なので、ダイ

オード 1N4148を入れました。このダイオードは安易に応答の遅い整流用を使わない方が良

いです。合わないものを使うとサージを吸収できずにトランジスタを壊したり、リレーを

動かす度にリセットがかかる回路ができあがります。 

 作動電圧 5Vの 941H-2C-5Dであれば 9Vの ACアダプタは不要です。この場合、駆動には

5V、約 30mAが必要で、リレーの電源は 5V端子から取りましょう。Arduino UNOのピンの電

流出力はカタログスペックでは最大 40mAとされており、一見すると直接つないで動かせそ

うですが、実際には出力電流が増えると出力電圧が低下します。こちらで実験したところ、

UNOのデジタル出力の電圧は何もデバイスをつながないときは 4.8V程度ですが、15mAで

4.4V、30mAで 4V、40mAで 3.6Vでした。そのため、30mA出力すると 941H-2C-5Dを駆動す

る電圧が確保できなくなる可能性があります。たとえ机上で動いても、部品の個体差で動

かなくなったり、温度変化で動かなくなったりしそうなので、ギリギリの設計はやめてト

ランジスタなどを入れるのをお勧めします。 

 リレーは機械式の接点を動かすという一見すると分かりやすそうなものですが、駆動に

必要な電流量が微妙に多かったり、大電流を流すと接点にスパークが生じたり、接点にチ

ャタリングが発生したりと扱いにくい場面があります。一方、ソリッドステート・リレー

(SSR)と呼ばれる半導体式のリレーが存在し、より簡単に大電流を制御できます。これを使

えば Arduinoの出力から AC100Vすら制御可能です。ゼロクロス機能を持つ SSRは AC専用

で、自動的に交流の電圧が 0Vになった瞬間を捉えて電源を ON/OFFするので、電気的なノ

イズが発生せず周辺の回路に対する影響も抑えられます。例えば SSRキット 35Aや

S108T02(SSRキット 8A)などです。ただし、SSRは通電中に発熱しますので、カタログスペ

ックは大電流に対応していても放熱しないと十分な性能を発揮できないことがあります。

また、AC100Vの取り扱いについては当然リスクが伴いますので、ここで取り扱う回路はデ

バイスのデータシート等を参照して注意点を確認した後、然るべき知識を持った方のみが

自己責任の上で挑戦して下さい。 

 

http://akizukidenshi.com/catalog/g/gP-02004/
http://akizukidenshi.com/catalog/g/gI-04268/
http://akizukidenshi.com/catalog/g/gI-00941/
http://akizukidenshi.com/catalog/g/gP-01229/
http://akizukidenshi.com/catalog/g/gK-00210/
http://akizukidenshi.com/catalog/g/gI-05568/
http://akizukidenshi.com/catalog/g/gK-06009/

SSRを用いた回路の例 

(注意：AC100Vの取り扱いにはリスクが伴いますので然るべき知識を持った方のみが自己責任の

上で挑戦して下さい) 

 

 

（西日本農業研究センター 黒崎秀仁） 

 


